
Searching for Resolvable Block Designs with
PROC OPTEX

Introduction
A design is resolvable when multiple blocks can be combined in such a way that each treatment occurs
equally often in each collection of blocks. If each treatment occurs ˛ times in each such collection of
blocks, then each collection is said to contain an ˛-replicate of the treatments, or simply a replicate if ˛ D 1.
Resolvability is a desirable feature of a design because it ensures orthogonality between treatments and
nuisance factors that might be of concern. For example, resolvability in sequential experimentation, where
replicates correspond to time periods, is used to mitigate time effects. Resolvability can likewise be useful in
multisite experiments and in experiments where multiple individuals handle experimental runs (Morgan and
Reck 2007). Resolvability is often a critical feature of how an experiment is performed, perhaps because
treatments are available in equal-sized replicate batches that must be used up before another batch can be
supplied.

The SAS source code for this example is available as an attachment in a text file. In Adobe Acrobat,
right-click the icon in the margin and select Save Embedded File to Disk. You can also double-click to
open the file immediately.

All You Really Need to Know
Suppose you are a food researcher who is studying wine preferences. You want your panel of 20 wine experts
to conduct taste tests of 50 varieties of wine. Obviously, they cannot taste all 50 wines in the same session, at
least not if you want to trust the ratings of the last dozen or so. So you propose to present the wines to them
in three sessions of 17, 17, and 16 wines apiece. You have two goals for the design:

1. Each expert should judge each wine once.

2. Each pair of wines should occur in the same session in as balanced a way as possible.

The following SAS statements invoke the OPTEX procedure from SAS/QC software to construct a good
experimental design for this study. You create two data sets that, respectively, describe the wine varieties
that you are studying and the expert/session setup that you have decided on. The OPTEX procedure’s
syntax describes the roles of these data sets in defining the “treatment” and “block” models, respectively,
and the PRIOR= option in the second MODEL statement is the essential feature that enables the resulting
arrangement to satisfy the two goals. The %RDBEval macro, which is documented in “Appendix: The
%RBDEval Macro Syntax” on page 9, evaluates the resulting design for efficiency with respect to replicates


/*--------------------------------------------------------------------*/
/* Copyright (c) 2014 SAS Institute Inc. Cary NC 27512  USA           */
/*                                                                    */
/*    NAME: RBDEVAL                                                   */
/* PURPOSE: Evaluate treatment designs with nested blocks             */
/* SUPPORT: Randy.Tobias@SAS.com          UPDATE: July 11, 2014       */
/*                                                                    */
/*--------------------------------------------------------------------*/
/* INPUT:                                                             */
/*    Design - name of data set with the design to be evaluated    .  */
/*    vName  - name of treatment or variety variable                  */
/*    rName  - name of replicate or first-level block variable        */
/*    bName  - name of blocks-within-replicate or second-level        */
/*             block variable                                         */
/*    nCheck - non-nested check designs, 0 by default                 */
/*                                                                    */
/* ALGORITHM:                                                         */
/* 1) Levels and joint levels are counted to determine whether the    */
/*    block structure is balanced and orthogonal.                     */
/* 2) OPTEX is used to compute the D- and A-efficiency of the design  */
/*    with respect to first- and second-level blocking only,          */
/*    respectively.  For balanced block structures, the block design  */
/*    D-efficiency for each situation is also computed.               */
/* 3) If a non-zero value for NCHECK= is specified, OPTEX is used to  */
/*    find the best design for each of these two situations, with     */
/*    ITER=&NCHECK, and the ratios of the design's efficiency         */
/*    measures to those of these best designs are computed.           */
/* 4) Finally, the resulting efficiencies are displayed---just the    */
/*    the raw ones computed in Step (2) by default, or additionally   */
/*    the efficiencies relative to OPTEX if NCHECK= is specified.     */
/*                                                                    */
/* Note that both the block design efficiency computed for balanced   */
/* block structures in (2) and the relative efficiencies computed     */
/* optionally in (3) get at the question of how good the design is    */
/* with what you can truly achieve given the blocking structure. The  */
/* block design efficiency is the ratio of the D criterion for the    */
/* design to that of a balanced incomplete block design, if it        */
/* exists.                                                            */
/*--------------------------------------------------------------------*/

%macro RBDEval(Design, vName, rName, bName, nCheck=0);
%local _nopt; %let _nopt = %sysfunc(getoption(notes));
options nonotes;

ods select none;

/*
/  Determine number of levels of variables and evaluate whether the
/  block structure is balanced and orthogonal.
/---------------------------------------------------------------------*/
   proc freq data=&Design noprint;
      table &vName        / out=_fV;
      table &rName        / out=_fR;
      table &rName*&bName / out=_fRB;
   run;

   proc summary data=_fV;
      var COUNT;
      output out=_vNum N=vNum;
   proc summary data=_fR;
      var COUNT;
      output out=_rNum N=rNum   std=rStd;
   proc summary data=_fRB;
      var COUNT;
      output out=_rbNum N=rbNum std=rbStd mean=rbAvg;
   run;

   data _null_; set &Design;
      call symput('N'   ,trim(left(_N_  )));
   data _null_; set _vNum;
      call symput('vNum',trim(left(vNum)));
   data _null_; set _rNum;
      call symput('rNum',trim(left(rNum)));
      call symput('rStd',trim(left(rStd)));
      call symput('rBal',trim(left(rStd < 1e-8)));
   data _null_; set _rbNum;
      call symput('rbNum',trim(left(rbNum)));
      call symput('rbStd',trim(left(rbStd)));
      call symput('rbBal',trim(left(rbStd < 1e-8)));
      call symput('rbAvg',trim(left(rbAvg)));
   run;

   %let BalBlock = %eval(&rBal & &rbBal);

   %if (&BalBlock) %then %do;
      %let pNum = &rbAvg;
      %let bNum = %eval(&rbNum / &rNum);

      %put Block structure: (&rNum * &bNum * &pNum);
      %end;

/*
/  Create the variety candidate set, the block design, and a candidate
/  set for variety and replicate only,
/---------------------------------------------------------------------*/
   data RBDTmtCan;
      do &vName = 1 to &vNum;
         output;
         end;
   data RBDBlkDesign; set &Design;
      keep &rName &bName;
   data RBDRepTmtCan;
      do &vName = 1 to &vNum;
      do &rName = 1 to &rNum;
         output;
         end; end;
   run;

   %let l1 = %length(&rName);
   %let l2 = %length(&bName);
   %let lEval = %eval(&l1 + &l2 + 2);


/*
/  Collect design's efficiency for
/     1) just treatments and replicates, ignoring blocks-within-
/        replicates, and
/     2) just treatments and blocks, ignoring replicates.
/  If block structure is balanced, use the appropriate syntax for
/  each, in order to get the BD efficiency; otherwise, just use the
/  the block structure data set.
/---------------------------------------------------------------------*/
   data Eff; if (0); run;
   %if (&BalBlock) %then %do;
      proc optex data=RBDTmtCan coding=orthcan;
         class &vName;
         model &vName;
         generate initdesign=&Design method=sequential;
         block structure=(&rNum)%eval(&bNum*&pNum) init=chain noexchange niter=0;
         ods output BlockDesignEfficiencies=iEff;
      quit;

      data iEff; set iEff;
         length Evaluation $ &lEval;
         Evaluation = "&rName";
      data Eff; set Eff iEff;
      run;

      proc optex data=RBDTmtCan coding=orthcan;
         class &vName;
         model &vName;
         generate initdesign=&Design method=sequential;
         block structure=(%eval(&rNum*&bNum))&pNum init=chain noexchange niter=0;
         ods output BlockDesignEfficiencies=iEff;
      quit;

      data iEff; set iEff;
         length Evaluation $ &lEval;
         Evaluation = "&bName(&rName)";
      data Eff; set Eff iEff;
      run;

      %let last = BCriterion;
      %let ASpace = 1;
      %end;
   %else %do;
      proc optex data=RBDTmtCan coding=orthcan;
         class &vName;
         model &vName;
         generate initdesign=&Design method=sequential;
         block design=RBDBlkDesign init=chain noexchange niter=0;
         class &rName;
         model &rName;
         ods output BlockDesignEfficiencies=iEff;
      quit;

      data iEff; set iEff;
         length Evaluation $ &lEval;
         Evaluation = "&rName";
      data Eff; set Eff iEff;
      run;

      proc optex data=RBDTmtCan coding=orthcan;
         class &vName;
         model &vName;
         generate initdesign=&Design method=sequential;
         block design=RBDBlkDesign init=chain noexchange niter=0;
         class &rName &bName;
         model &rName &bName &rName*&bName;
         ods output BlockDesignEfficiencies=iEff;
      quit;

      data iEff; set iEff;
         length Evaluation $ &lEval;
         Evaluation = "&bName(&rName)";
      data Eff; set Eff iEff;
      run;

      %let last = ACriterion;
      %let ASpace = 4;
      %end;

/*
/  If optimal checking is requested, go ahead and find the designs
/  that are optimal with respect to the replicate and blocks-within-
/  replicates models used previously for evaluation, and compute the
/  given designs efficiencies relative to those.
/---------------------------------------------------------------------*/
   %if (&nCheck) %then %do;
      data Check; if (0); run;

      %if (&BalBlock) %then %do;
         proc optex data=RBDTmtCan coding=orthcan;
            class &vName;
            model &vName;
            block structure=(&rNum)%eval(&bNum*&pNum) niter=&nCheck keep=1;
            ods output BlockDesignEfficiencies=iCheck;
         quit;
         data Check; set Check iCheck;
         run;

         proc optex data=RBDTmtCan coding=orthcan;
            class &vName;
            model &vName;
            generate initdesign=&Design method=sequential;
            block structure=(%eval(&rNum*&bNum))&pNum niter=&nCheck keep=1;
            ods output BlockDesignEfficiencies=iCheck;
         quit;
         data Check; set Check iCheck;
         run;

         data Eff; merge Eff Check(rename=(DCriterion=CheckDCriterion
                                           ACriterion=CheckACriterion
                                           BCriterion=CheckBCriterion));
            rDCriterion = 100*DCriterion/CheckDCriterion;
            rACriterion = 100*ACriterion/CheckACriterion;
            rBCriterion = 100*BCriterion/CheckBCriterion;
            format rDCriterion rACriterion rBCriterion 8.4;
         run;
         %end;
      %else %do;
         proc optex data=RBDTmtCan coding=orthcan;
            class &vName;
            model &vName;
            block design=RBDBlkDesign niter=&nCheck keep=1;
            class &rName;
            model &rName;
            ods output BlockDesignEfficiencies=iCheck;
         quit;
         data Check; set Check iCheck;
         run;

         proc optex data=RBDTmtCan coding=orthcan;
            class &vName;
            model &vName;
            block design=RBDBlkDesign niter=&nCheck keep=1;
            class &rName &bName;
            model &rName &bName &rName*&bName;
            ods output BlockDesignEfficiencies=iCheck;
         quit;
         data Check; set Check iCheck;
         run;

         data Eff; merge Eff Check(rename=(DCriterion=CheckDCriterion
                                           ACriterion=CheckACriterion));
            rDCriterion = 100*DCriterion/CheckDCriterion;
            rACriterion = 100*ACriterion/CheckACriterion;
            format rDCriterion rACriterion 8.4;
         run;
         %end;

      %end;

ods select all;

proc template;
   define table RBDEval;

      column Evaluation  DCriterion  ACriterion  BCriterion
                        rDCriterion rACriterion rBCriterion;

      define header h;
         text "&vname Efficiency for &rName and &bName-within-&rName";
         space=1;
      end;

      define Evaluation;
         id header="Evaluation" style=RowHeader;
      end;

      define DCriterion;
         header="D" format=8.4 space=1;
      end;

      define ACriterion;
         header="A" format=8.4 space=&ASpace;
      end;

      define BCriterion;
         header="BD" format=8.4 space=4;
      end;

      %if (&nCheck) %then %do;
         define header hRaw;
            text "Raw";
            start=DCriterion end=&Last expand='-';
         end;
     %end;

      define rDCriterion;
         header="D" format=8.4 space=1;
      end;

      define rACriterion;
         header="A" format=8.4 space=&ASpace;
      end;

      define rBCriterion;
         header="BD" format=8.4;
      end;

      define header hRel;
         text "Relative";
         start=rDCriterion end=r&Last expand='-';
      end;

   end;
run;

proc sgrender data=Eff template="RBDEval";
run;

options &_nopt;
%mend;

data Wines;
   do Wine = 1 to 50;
      output;
   end;
run;

data Setup;
   do Subject = 1 to 20;
      do Session = 1 to  3;
         if (Session < 3) then nPlot = 17;
         else                  nPlot = 16;
         do Plot = 1 to nPlot;
            output;
         end;
      end; 
   end;
   drop nPlot;
run;

proc optex data=Wines coding=orthcan seed=16899;
   class Wine;
   model Wine;
   block design=Setup;
   class Subject Session;
   model Subject, Session(Subject) / prior=0,10;
   output out=Design;
run;

%RBDEval(Design,Wine,Subject,Session,nCheck=10);

data Candidates;
   do Treatment = 1 to 4;
      output;
   end;
run;

data Design;
   do Replicate = 1 to 3;
      do Block = 1 to 2;
         do Plot = 1 to 2;
            input Treatment @@;
            output;
         end;
      end;
   end;
datalines;
1 2   3 4
1 3   2 4
1 4   2 3
;

%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);

   
data Candidates;
   do Treatment = 1 to 15;
      output;
   end;
run;

data BlockStructure;
   do Replicate = 1 to 7;
      do Block = 1 to 5;
         do Plot = 1 to 3;
            output;
         end;
      end;
   end;
run;

proc optex data=Candidates coding=orthcan seed=492069001;
   class Treatment;
   model Treatment;
   blocks design=BlockStructure;
   class Replicate Block;
   model Replicate Block(Replicate);
   output out=Design;
   ods select BlockDesignEfficiencies;
run;


%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);

proc freq data=Design;
   table Treatment*Replicate / norow nocol nopct nocum;
run;

proc optex data=Candidates coding=orthcan seed=607104001;
   class Treatment;
   model Treatment;
   blocks design=BlockStructure niter=10000 keep=10;
   class Replicate Block;
   model Replicate, Block(Replicate) / prior=0,100;
   output out=Design;
   ods select BlockDesignEfficiencies;
run;

%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);

data Candidates;
   do Treatment = 1 to 4;
      output;
   end;
run;

data Design;
   do Replicate = 1 to 3;
      do Block = 1 to 2;
         do Plot = 1 to 2;
            input Treatment @@;
            output;
         end;
      end;
   end;
datalines;
1 2   3 4
1 3   2 4
1 4   2 3
;

%RBDEval(Design,Treatment,Replicate,Block);

%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);



SAS source code for this example. Right-click to save file.

and blocks within replicates. A D-efficiency rating of 100% with respect to replicates indicates that the
design is resolvable. The D-efficiency rating with respect to blocks within replicates indicates how well
balanced the design is (see the section “Optimality Criteria” in the chapter “The OPTEX Procedure” of the
SAS/QC User’s Guide for a discussion of the D-optimality criteria).

data Wines;
do Wine = 1 to 50;

output;
end;

run;

data Setup;
do Subject = 1 to 20;

do Session = 1 to 3;
if (Session < 3) then nPlot = 17;
else nPlot = 16;
do Plot = 1 to nPlot;

output;
end;

end;
end;
drop nPlot;

run;

proc optex data=Wines coding=orthcan seed=16899;
class Wine;
model Wine;
block design=Setup;
class Subject Session;
model Subject, Session(Subject) / prior=0,10;
output out=Design;

run;

%RBDEval(Design,Wine,Subject,Session,nCheck=10);

Figure 1 shows that the best design found has a treatment D-efficiency of 96.58.

Figure 1 Design Criteria for Replicates and Blocks within Replicates

The OPTEX ProcedureThe OPTEX Procedure

Design
Number

Treatment
D-Efficiency

Treatment
A-Efficiency

1 96.5760 96.5530

2 96.5759 96.5529

3 96.5756 96.5523

4 96.5756 96.5521

5 96.5751 96.5513

6 96.5750 96.5510

7 96.5749 96.5508

8 96.5748 96.5506

9 96.5745 96.5501

10 96.5744 96.5498

Figure 2 shows that the resulting design has a block design D-efficiency of 100%, so the design is resolvable
and satisfies the first goal precisely. It also shows that the design has a raw block-within-replicates D-
efficiency of 95.88 and a relative block-within-replicates D-efficiency of 99.98. The relative efficiency
measures compare the raw efficiency measures to the efficiency measures from the best design that PROC
OPTEX can find by using the number of search iterations that are specified in the nCheck= argument. The
first row compares the raw measures to those for the best design found that contains replicates only. The
second row compares the raw measures to the best design found that contains blocks only. By construction,
the design in this example cannot be completely balanced, so the second goal can be only approximately
satisfied. However, PROC OPTEX finds a design that is fairly close to being balanced. Each pair of wines is
tasted together between 23 and 30 times, with most pairs being tasted together 26–27 times.

Figure 2 Design Efficiency Evaluation

Wine Efficiency for Subject and Session-within-Subject

Raw Relative

Evaluation D A D A

Subject 100.0000 100.0000 100.0000 100.0000

Session(Subject) 95.8849 95.8516 99.9773 99.9548

The remainder of this paper reviews the OPTEX procedure and explains why this method works.

Resolvable Designs
Table 1 shows a resolvable block design for four treatments in six blocks of size 2; the pairs of blocks in each
row of the table make up a single replicate of the treatments. Because the blocks of this design consist of all
pairs of treatments, it is also balanced, and therefore it is a resolvable balanced incomplete block design.

Table 1 A Resolvable Balanced Incomplete Block Design

Block
Replicate 1 2

1 [1 2] [3 4]
2 [1 3] [2 4]
3 [1 4] [2 3]

Resolvable balanced incomplete block designs are D-optimal with respect to two different kinds of blocks:
the primary blocks and the replicates, considered as blocks. To confirm this, the following statements use the
%RBDEval macro to evaluate the arrangement of treatments that is shown in Table 1, both as a design for six
blocks of size 2 and as a design for three blocks of size 4.

The following DATA steps create two data sets: Candidates and Design. Candidates contains the candidate
points for the design, and Design contains a set of points for the design that is to be evaluated for D-efficiency.

data Candidates;
do Treatment = 1 to 4;

output;
end;

run;

data Design;
do Replicate = 1 to 3;

do Block = 1 to 2;
do Plot = 1 to 2;

input Treatment @@;
output;

end;
end;

end;
datalines;
1 2 3 4
1 3 2 4
1 4 2 3
;

%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);

Figure 3 displays the results.

Figure 3 Design Efficiency Evaluation

Treatment Efficiency for Replicate and Block-within-Replicate

Raw Relative

Evaluation D A BD D A BD

Replicate 100.0000 100.0000 100.0000 100.0000 100.0000 100.0000

Block(Replicate) 66.6667 66.6667 100.0000 100.0000 100.0000 100.0000

As claimed, the design has 100% block design D-efficiency with respect to both block structures. The design
also has 100% treatment D-efficiency with respect to replicates because each replicate contains each treatment
once.

Searching for a Resolvable Design
You can use PROC OPTEX to search for good resolvable block designs, but doing so requires a little trick.
The DESIGN= option in the BLOCK statement enables you to specify a block structure that involves both
replicates and blocks within replicates, but this is not enough. To see why, consider the following PROC
OPTEX example, which sets up a candidate set of 15 treatments and a block structure data set that consists
of seven replicates of five blocks of size 3, and then uses PROC OPTEX to find the best way to assign these
15 treatments to such a block structure, considering effects for both replicates and block within replicates.

The following SAS statements create two data sets, Candidates and BlockStructure. Candidates contains the
candidate points for the design, and BlockStructure contains a block structure that involves both replicates
and blocks within replicates.

data Candidates;
do Treatment = 1 to 15;

output;
end;

run;

data BlockStructure;
do Replicate = 1 to 7;

do Block = 1 to 5;
do Plot = 1 to 3;

output;
end;

end;
end;

run;

The following statements use PROC OPTEX to search for a good design for the candidate points that
are stored in the data set Candidates by using the blocking structure that is contained in the data set
BlockStructure:

proc optex data=Candidates coding=orthcan seed=492069001;
class Treatment;
model Treatment;
blocks design=BlockStructure;
class Replicate Block;
model Replicate Block(Replicate);
output out=Design;
ods select BlockDesignEfficiencies;

run;

The DATA= option in the PROC OPTEX statement specifies that the data set Candidates contains the
candidate points for the design. The SEED= option specifies an integer to use to start the pseudorandom
number generator for initialization and guarantees reproducibility. The first CLASS statement refers to
the data set Candidates and specifies Treatment as a classification variable. The first MODEL statement
represents the treatment model; it specifies that the model consists of the main treatment effects. The
DESIGN= option in the BLOCKS statement specifies that BlockStructure contains the fixed covariates for the
model. In this case, the covariates are the replicates and the blocks. The CLASS statement that follows the
BLOCKS statement refers to the data set BlockStructure and the model for the fixed covariates and specifies
Replicate and Block as classification variables. Similarly, the MODEL statement that follows the BLOCKS
statement refers to the data set BlockStructure and the model for the fixed covariates; it specifies fixed effects
for Replicate and Block nested within Replicate. The OUTPUT statement saves the best design in the data
set Design, which is specified in the OUT= option.

Figure 4 shows that the best design found has a treatment D-efficiency of 71.43.



6 F

Figure 4 Design Criteria for Replicates and Blocks within Replicates

The OPTEX ProcedureThe OPTEX Procedure

Design
Number

Treatment
D-Efficiency

Treatment
A-Efficiency

1 71.4286 71.4286

2 71.4286 71.4286

3 71.4286 71.4286

4 71.3371 71.2444

5 71.3371 71.2444

6 71.3371 71.2444

7 71.3371 71.2444

8 71.3371 71.2444

9 71.3371 71.2444

10 71.2004 70.9710

The following statement calls the %RBDEval macro, which evaluates the design:

%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);

The resulting efficiency measures for the two different blocking structures are shown in Figure 5. The design
has a block design D-efficiency with respect to replicates of 97.3. The design is not resolvable because it is
not 100% efficient with respect to replicates. The design has a relative block design D-efficiency with respect
to blocks within replicates of 100%, indicating that the design achieves balance with respect to blocks within
replicates.

Figure 5 Design Criteria for Replicates and Blocks within Replicates

Treatment Efficiency for Replicate and Block-within-Replicate

Raw Relative

Evaluation D A BD D A BD

Replicate 97.3299 97.2418 97.3299 97.3299 97.2418 97.3299

Block(Replicate) 71.4286 71.4286 100.0000 100.0000 100.0000 100.0000

The following PROC FREQ step confirms that some treatments occur twice in some replicates and other
treatments do not occur at all:

proc freq data=Design;
table Treatment*Replicate / norow nocol nopct nocum;

run;

The resulting coincidence counts are shown in Figure 6.

Figure 6 Treatment-by-Replicate Coincidence Counts for Optimal Figure 4 Design

The FREQ ProcedureThe FREQ Procedure

Frequency Table of Treatment by Replicate

Treatment

Replicate

1 2 3 4 5 6 7 Total

1 1 0 0 2 2 1 1 7

2 2 1 1 0 1 1 1 7

3 1 1 0 2 1 1 1 7

4 0 2 1 1 1 1 1 7

5 1 0 2 1 0 0 3 7

6 2 1 2 0 1 1 0 7

7 1 1 1 0 2 1 1 7

8 0 2 1 1 1 1 1 7

9 1 1 1 1 1 1 1 7

10 2 0 1 1 1 1 1 7

11 0 2 1 1 1 1 1 7

12 2 1 1 1 1 1 0 7

13 0 1 2 1 1 1 1 7

14 1 1 1 1 0 2 1 7

15 1 1 0 2 1 1 1 7

Total 15 15 15 15 15 15 15 105

The reason why this optimal blocking approach fails to find a resolvable design is that PROC OPTEX can
optimize only one definition of block D-efficiency, which is jX 0AX j, at a time, and the blocking model that is
used here defines the matrix A based only on the effect of blocks within replicates. Call this matrix AB , and
call the corresponding matrix for replicates considered as blocks AR. Technically, AR and AB are defined as
the projectors onto the residual spaces for the models with effects only for replicates and only for blocks
within replicates, respectively. To find a design that is both resolvable and balanced, you need to maximize
the determinant of an information matrix that combines these two separate information matrices:

D˛ D j˛X 0ARX C .1 � ˛/X
0ABX j; 0 < ˛ < 1

It can be shown that if a resolvable design exists, it maximizes such a criterion.

This is where the trick comes in. The PRIOR= option in the MODEL statement can be used with the
BLOCKS model and with the treatment model. For a block design whose blocks and replicates are of equal
sizes, a block model of the form

class Replicate Block;

model Replicate, Block(Replicate) / prior=0,�;

defines a block D-efficiency of the form shown earlier, where

˛ D
�

N C �

where N is the size of the design. That is, a resolvable block design is Bayes optimal in the sense of
DuMouchel and Jones (1994). The intuitive interpretation of this result is that, by claiming a certain amount
of prior information about blocks within replicates, you are freeing PROC OPTEX to try to find a design that
contains information about replicates, too.

The following statements exploit this method to find a resolvable balanced incomplete block design for 15
treatments in seven replicates of five blocks of size 3, by claiming about 100 observations’ worth of prior
information about the blocks-within-replicates effect:

proc optex data=Candidates coding=orthcan seed=607104001;
class Treatment;
model Treatment;
blocks design=BlockStructure niter=10000 keep=10;
class Replicate Block;
model Replicate, Block(Replicate) / prior=0,100;
output out=Design;
ods select BlockDesignEfficiencies;

run;

%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);

Notice how high the NITER= option value is set in the BLOCKS statement. Combinatorial designs of any
sort can be difficult for PROC OPTEX to find, so the more iterations you allow, the better your odds of
finding the theoretical optimum. On the other hand, PROC OPTEX invariably finds a pretty good design,
if not the very best possible design. When you use the PRIOR= option, you must use commas to separate
groups of effects in the MODEL statement that have the same prior precision. This example has two groups:
Replicate and Block(Replicate). The PRIOR= option specifies a prior precision value of 0 for Replicate and
a prior precision value of 100 for Block(Replicate).

Figure 7 shows that the best design found by using the prior information has a treatment D-efficiency rating
of 96.27.

Figure 7 Design Criteria for Bayesian Design

The OPTEX ProcedureThe OPTEX Procedure

Design
Number

Treatment
D-Efficiency

Treatment
A-Efficiency

1 96.2733 96.2733

2 96.2733 96.2733

3 96.2716 96.2699

4 96.2710 96.2687

5 96.2710 96.2687

6 96.2710 96.2687

7 96.2710 96.2687

8 96.2710 96.2687

9 96.2710 96.2687

10 96.2710 96.2687

Figure 8 shows that the design achieves a 100% block design D-efficiency with respect to replicates, indicating
that the design is resolvable. The design also achieves a 100% block design D-efficiency with respect to
blocks within replicates, indicating that the design is balanced.

Figure 8 Optimal Figure 8 Design Evaluated for Replicates

Treatment Efficiency for Replicate and Block-within-Replicate

Raw Relative

Evaluation D A BD D A BD

Replicate 100.0000 100.0000 100.0000 100.0000 100.0000 100.0000

Block(Replicate) 71.4286 71.4286 100.0000 100.0000 100.0000 100.0000

A question that naturally arises when you use this approach is, What value of � should you use for the
nonzero prior on blocks within replicates? Mathematically, a resolvable design is optimal for any value of � ,
but if you set it too small or too large, one or the other of the component information matrices will dominate
the efficiency criterion. It doesn’t seem possible to give any definitive advice on this point, other than to try
several values. You want a prior high enough to ensure resolvability, but not so high that PROC OPTEX fails
to find the most efficient resolvable design for the situation. You might, for example, start with � D N=10,
one tenth the size of the design. If the resulting design is not resolvable, increase � to, for example, C �N=10,
where 2 � C � 10. When PROC OPTEX finds a resolvable design, it is usually highly efficient compared to
the most efficient possible resolvable design for the specific situation. Nevertheless, you can try reducing �
by, for example, half, to see whether PROC OPTEX can find a resolvable design that has higher efficiency.

Contributors
Cliff Pereira and Luna Sun of Oregon State University provided the applications that motivated this work and
a practical review of the methodology. Randy Tobias, the developer of PROC OPTEX and the %RDBEval
macro, provided the optimal design expertise.

Appendix: The %RBDEval Macro Syntax
The %RDBEval macro evaluates an experimental design for efficiency with respect to replicates and blocks
within replicates by using PROC OPTEX. When calling %RBDEval, you must specify a SAS data set that
contains the design to be evaluated, and three variables that identify the treatments, replicates, and blocks,
respectively.

%RBDEval Macro Syntax
%RBDEval(Design, vName, rName, bName, nCheck=)

Required Arguments

Design SAS-data-set
specifies a SAS data set that contains the design.

vName variable
specifies the variable that indexes the treatments.

rName variable
specifies the variable that indexes the replicates.

bName variable < number >
specifies the variable that indexes the blocks.

Optional Argument

nCheck=number
requests a search for optimal replicate-only and blocks-only designs, respectively, for the purpose of
computing the relative efficiencies of a design; and specifies the number of times to repeat the search
from different initial designs. The default value is 0, which indicates that no search is to be performed
and that the relative efficiencies are not to be computed.

The following example uses the design from Table 1 and demonstrates how to use the %RBDEval macro,
first omitting the nCheck= argument, and then by specifying the nCheck= argument:

data Design;
do Replicate = 1 to 3;

do Block = 1 to 2;
do Plot = 1 to 2;

input Treatment @@;
output;

end;
end;

end;
datalines;
1 2 3 4
1 3 2 4
1 4 2 3
;

%RBDEval(Design,Treatment,Replicate,Block);

Figure 9 shows that when you do not specify the nCheck= argument, %RBDEval displays only the raw
efficiency measures and, if appropriate, the block design efficiency measures. The block design measures are
displayed if, as in this case, the replicates and blocks are balanced, meaning that there are an equal number of
blocks in every replicate and an equal number of plots in every block. The block design efficiency measures
compare the raw D-efficiency measures to those that optimal balanced designs would theoretically have, if
they exist.

Figure 9 Design Efficiency Evaluation

Treatment Efficiency for Replicate and
Block-within-Replicate

Evaluation D A BD

Replicate 100.0000 100.0000 100.0000

Block(Replicate) 66.6667 66.6667 100.0000

The following statement calls the %RBDEval macro and specifies the nCheck= argument:

%RBDEval(Design,Treatment,Replicate,Block,nCheck=10);

Figure 10 shows that when you do specify the nCheck= argument (with a nonzero value for the argument), the
%RBDEval macro displays both the raw and relative efficiency measures. The relative efficiency measures
compare the raw efficiency measures to the efficiency measures from the best designs that PROC OPTEX
can find by using the number of search iterations that you specify in the nCheck= argument. The first row
compares the raw measures to those for the best design found that contains replicates only. The second row
compares the raw measures to the best design found that contains blocks only.

Figure 10 Design Efficiency Evaluation

Treatment Efficiency for Replicate and Block-within-Replicate

Raw Relative

Evaluation D A BD D A BD

Replicate 100.0000 100.0000 100.0000 100.0000 100.0000 100.0000

Block(Replicate) 66.6667 66.6667 100.0000 100.0000 100.0000 100.0000

